

The Mjølner BETA System

Container Libraries

Reference Manual

Mjølner Informatics Report

MIA 92-22(1.3)

August 1996

Copyright © 1992-96 Mjølner Informatics ApS.
All rights reserved.
No part of this document may be copied or distributed
without prior written permission from Mjølner Informatics


Contents

THE CONTAINER LIBRARIES.....	1
CONTAINER.....	1
COLLECTION.....	2
MULTISET.....	2
SET.....	3
CLASSIFICATIONSET.....	3
HASHTABLE.....	5
EXTENSIBLEHASHTABLE.....	5
ARRAYCONTAINER.....	5
SEQUENTIALCONTAINER.....	6
STACK.....	6
QUEUE.....	6
DEQUE.....	6
PRIOQUEUE.....	6
LIST.....	7
RECLIST.....	8
USING THE CONTAINER LIBRARIES.....	9
EXAMPLES OF USE OF THE CONTAINER LIBRARIES.....	10
INTERFACE DESCRIPTION FOR THE CONTAINER LIBRARIES.....	19
CONTAINER.BET.....	19
COLLECTION.BET.....	22
SETS.BET.....	24
CLASSIFICATION.BET.....	26
ARRAYCONTAINER.BET.....	27
HASHTABLE.BET.....	29
SEQCONTAINERS.BET.....	32
LIST.BET.....	36
RECLIST.BET.....	41
INDEX.....	43

The Container Libraries

This document describes version 1.5 of the container libraries. The container libraries implements many fairly often used data structures, such as sets, lists, stacks, etc., giving solutions to a very broad range of data structuring issues. It is intended that the container libraries will be extended in the future to include data structures such as trees, and mappings. The container libraries contains a number of abstract patterns which can be used as the basis for other container data structures.

The entire inheritance hierarchy for the container libraries is:



The container libraries are fragmented into relatively small fragments to enable the programmers to be very selective in picking the proper data structures without being burdened with the overhead of the other data structures.

We will in the following discuss the patterns in the container libraries individually.

Container

`Container` is the abstract superpattern for all patterns in the container libraries, defining all common attributes of container patterns.

Container patterns are generic patterns in the sense that the element type of the elements kept in the container is specified through the `element` virtual attribute.

`Container` defines the following additional attributes:

`init`: To be invoked before the container object may be used.

`clear`: empties the container.

`empty`: returns true, if the container does not contain any elements.

`size`: returns the number of elements in the container.

`capacity`: returns the current capacity of the container. Most containers will dynamically expand as usage demands, and `capacity` returns the current capacity

Container attributes

(subject to changes later, when usage demands). If the capacity of the container is in principle indefinite, capacity returns -1.

`has`: takes a reference to an element object, and returns true if the element is in the container.

`scan`: scans through the entire container, invoking `INNER` for each element in the container satisfying a predicate.

`find`: scans through the entire container, invoking `INNER` first time an element is found, satisfying a predicate. `find` also returns a reference to the element found.

`copy`: creates an exact copy of the container. Only elements satisfying a predicate is copied to the new container.

`equal`: a virtual procedure pattern, taking two element references. `Equal` returns true if the two element references are identical. `Equal` is used in the implementation of the above operations (and some of the later mentioned operations). If some other equality test is wanted for specific containers, this can be accomplished by further binding `equal` in that container.

Container also defines two exceptions:

`emptyContainerError`: is invoked if the container is empty, and some container operation is invoked which demands a non empty container.

`illegalCellReference`: is invoked, if some container operation is making illegal references internally in the container private data structures.

Collection

`Collection` is a subpattern of the `container` pattern and is the abstract superpattern for all position independent container patterns in the container libraries, defining all common attributes of these patterns.

`Collection` defines the following additional attributes:

`insert`: takes an element reference and inserts the element in the collection.

`delete`: takes an element reference and deletes that element from the collection.

`union`: takes a reference to another collection, and unifies the elements in both collections into `this(collection)`.

`diff`, `sect`, `symDiff`: similar to `union`.

Collection attributes

MultiSet

`MultiSet` is a subpattern of the `collection` pattern. `MultiSet` is an unstructured collection of element references, where duplicates are allowed. `MultiSet` does not define any additional operations.

Set

Set is a subpattern of the `multiSet` pattern. Set is like `multiSet`, except that duplicated are *not* allowed. Set does not define any additional operations.

ClassificationSet

ClassificationSet is a subpattern of the `set` pattern. ClassificationSet is used for dynamic classification of elements. ClassificationSet defines two additional attributes:

`subsets`: is a set containing the subsets of this `classificationSet`.


`insertSubset`: takes a `classificationSet` and inserts it as a subset of this `classificationSet`. The exception `illegalSubset` will be invoked if the subset inserted contains elements, that are not instances of a subpattern of the element type of this `classificationSet`.

`superSet`: if this `classificationSet` is a subset of another `classificationSet`, `superSet` will reference that `classificationSet`.

`scanUnclassified`: like `scan` except that it only scans those elements that are not member of any subsets of this `classificationSet`.

**ClassificationSet
attributes**


Let us illustrate by an abstract example, assuming that `css1` contains the elements a, b and c, `css2` contains the elements c, d and e, and finally `cs` contains the elements b and f:



Then after

```
css1[]->cs.insertSubset
```

`cs` will contain the original elements, and the `css1` subset:




Note that the element `b`, originally located in both `cs` and `css1`, now is located exclusively in `css1`. In this situation, `cs.scan(# ... #)` will run through the elements `a`, `b`, `c`, `d`, `e` and `f`. and `css1.scan(# ... #)` will run through the elements `b` and `f`.

If we then execute

```
css2[]->cs.insertSubset
```


we will have the following situation:



Note again, that `c` is exclusively in `css2`. If we now execute

```
g[]->css1.insert, f[]->css2.insert and b[]->cs.delete
```

we will have the following situation:



Finally,

```
css1.scanUnclassified(# ... #)
```

will only scan the elements `a` and `b`, whereas

```
css1.scan(# ... #)
```

will scan all elements (i.e. `a`, `b`, `c`, `d`, `e`, `f` and `g`).

HashTable

HashTable is a subpattern of the `collection` pattern. HashTable implements a hash table data structure with the following additional attributes:

`rangeInitial`: defines the range of internal hash values of the hash table.

`range`: returns the actual range of internal hash values of the hash table.

`hashFunction`: takes an element reference and returns a hash value for that element. In most uses of `hashTable`, this function must be further bound to give a more intelligent distribution of hash values.

`scanIndexed`: takes a hash value and scans all elements in the hash table with the same hash value.

`findIndexed`: takes a hash value and finds an element with the same hash value, satisfying predicate.

`statistics`: enables calculation of various statistics about the usage of the hash table.

**HashTable
attributes**

ExtensibleHashTable

ExtensibleHashTable is a subpattern of the `hashTable` pattern. Extensible-HashTable allows for extension of the range of hash values of the `hashTable`. Unless the specified `hashFunction` somehow explicitly can cope with dynamic changes to the hash range, the entire `hashTable` needs to be reshaped. ExtensibleHashTable implements the following additional attributes:

`extend`: extends the range of hash values of the hash table.

`rehash`: makes a total rehash of the contents of the hash table.

**Extensible-
HashTable
attributes**

ArrayContainer

ArrayContainer is a subpattern of the `container` pattern. ArrayContainer is an abstraction of an element repetition, implementing the following additional attributes:

`put`, `get`: inserts/returns an element reference at a given position.

`bubble-`, `shell-` and `quicksort`: sorting operations.

`capacityInitial`, `capacityIncrement`, `capacityExtend`: are used for controlling the strategy for extending the ArrayContainer. `capacityInitial` is used to specify the initial size of the repetition. `capacityIncrement` is used

**ArrayContainer
attributes**

to specify the additional index values to be allocated when the `capacityExtend` is invoked.

SequentialContainer

`SequentialContainer` is a subpattern of the `container` pattern and is the abstract superpattern for patterns in the container libraries that implements data structures, where the elements are sequentially ordered, and where only the elements in the front or back of this sequence can be manipulated (e.g. stacks and queues). `SequentialContainer` does not define any additional operations.

Stack

`Stack` is a subpattern of the `sequentialContainer` pattern. `Stack` implements an ordinary stack with `push`, `pop` and `top` operations.

Queue

`Queue` is a subpattern of the `sequentialContainer` pattern. `Queue` implements an ordinary queue with `insert`, `remove` and `front` operations.

Deque

`Deque` is a subpattern of the `sequentialContainer` pattern. `Deque` implements an ordinary double-ended queue with `insertFront`, `insertBack`, `removeFront`, `removeBack`, `front` and `back` operations.

PrioQueue

`PrioQueue` is a subpattern of the `sequentialContainer` pattern. `PrioQueue` implements a priority queue with `insert`, `remove`, `front` and `scanPriority` operations. All operations takes a number, identifying a priority. `ScanPriority` scans through all elements of a given priority.

List

List is a subpattern of the container pattern. It implements a double-chained list data structure. The list pattern is a little special, since many operations takes list positions as arguments instead of elements. The reason is, that the typical usage of lists are to manipulate the list itself (finding the element just before a given position in the list, etc.). This adds a little extra complexity to the operations of the list pattern, relative to the other container patterns.

Central to the understanding of the list pattern, is the understanding of the `theCellType` pattern. `TheCellType` is a virtual pattern, defined in container pattern (but it is a private attribute, except for the list subpattern). In the list pattern, `theCellType` contains the references to the successor and predecessor list positions, as well as the `elm` reference to the element object contained at this list position. `TheCellType` describes the list positions (also often called list elements).

List defines the following additional operations:

`prepend`: takes an element (i.e. a reference to an element) and adds it to the front of the list. `Prepend` returns the position of the element in the list (i.e. a reference to the instance of `theCellType`, containing the new element reference). This position may later be used to reference the element in the list (e.g. finding the `succ` of `pred` positions in the list).

List attributes

`append`: Similar to `prepend`, except that it inserts the element at the end of the list.

`head`: Returns the position in the beginning of the list.

`last`: Returns the position in the end of the list.

`tail`: Returns a copy of the list, except for the first list element which have been removed.

`preample`: Returns a copy of the list, except for the last list element which have been removed.

`concatenate`: takes two lists and returns a new list which is the concatenation of the two lists.

`insertBefore`, `insertAfter`: Takes an element and a list position, and inserts the element before (respectively after) the list position.

`delete`, `deleteBefore`, `deleteAfter`: takes a list position and deletes the list position (respectively the list position before or after) from the list.

`splitBefore`, `splitAfter`: takes a list position and returns two lists that are the lists from the beginning up to the list position, respectively from the list position until the end of the list. The list position will be in one of the lists.

`at`: takes an element and returns the first list position, containing a reference to that element.

`locate`: returns the position of the element in the list, satisfying predicate. This operation is similar to `find`, except that it returns the position, not the element.

`iterate`, `iterateFrom`, `iterateReverse`, `iterateReverseFrom`: iterates through the list structure in different ways. In contrast to `scan` (inherited from container), all `iterate` variants respects the sublist structure such that a sublist will *not* be traversed automatically.

`scanFrom`, `scanReverse`, `scanReverseFrom`: similar to the `iterate` operations above.

RecList

`RecList` is a subpattern of the list pattern, and makes it possible to represent recursive lists. `TheCellType` is further bound to include the `sublist` reference, and if the list position represents a sublist, `sublist` will contain a reference to the sublist. `TheCellType` further defines the `when` attribute, which may be used to specify actions to be executed in case the list position contains an element reference, respectively a sublist. Finally, `recList` defines the following additional operations:

RecList attributes `insertSublistBefore`, `insertSublistAfter`: Takes a list and a list position, and inserts the list as a sublist before (respectively after) the list position

`RecList` defines the exception `illegalSublist`, which is invoked if the elements of an sublist are not instances of a subpattern of the `element` type of this `recList`.

Using the Container Libraries

The container libraries consists of a number of fragments: `container`, `collection`, `sets`, `classification`, `arrayContainer`, `hashTable`, `seqContainers`, `list` and `recList`:

- `container.bet` contains the definition of the abstract pattern `container`. It is the origin fragment for all the other container fragments.
- `collection.bet` contains the definition of the abstract pattern `collection`. It is the origin for the `sets`, `classification` and `hashTable` fragments.
- `sets.bet` contains the definition of the `multiSet` and `set` patterns. It is the origin of the `classification` fragment.
- `classification.bet` contains the definition of the `classificationSet` pattern.
- `hashTable.bet` contains the definition of the `hashTable` and `extensible-HashTable` patterns.
- `arrayContainer.bet` contains the definition of the `ArrayContainer` pattern.
- `list.bet` contains the definition of the `list` pattern.
- `recList.bet` contains the definition of the `recList` pattern.
- `seqContainers.bet` contains the definition of the abstract pattern `sequentialContainer` along with the `stack`, `queue`, `prioQueue` and `deque` patterns.

An application may utilize the container libraries by including the appropriate library fragment. E.g. in order for an application to utilize the `multiSet` pattern, the application must have the following outline:

Using MultiSet

```

ORIGIN '~beta/basiclib/v1.5/betaenv';
INCLUDE '~beta/containers/v1.5/sets'
--- program: descriptor ---
(# ...
  recordBag: @multiSet(# element:: record #);
  aRecord: @record;
  aPerson: @person; (* person is a subpattern of record *)
  ...
do ...
  recordBag.init;
  ...
  aPerson[] -> recordBag.insert;
  ...
  aRecord[] -> recordBag.insert;
  ...
  recordBag.scan(# where:: (# current.key<3000 -> value #)
                 do current.print
                 #);
  (* prints all records in recordBag that has a key less than
   * 3000. It is assumed that record implements a virtual print
   * operation
   *)
  ...
#)

```

In order to utilize the other fragments, just replace

```
INCLUDE '~beta/containers/v1.5/sets'
```

with e.g.

```
INCLUDE '~beta/containers/v1.5/hashTable'
```

and in order to use more than one container fragment, just insert several INCLUDEs:

Using several containers

```
INCLUDE '~beta/containers/v1.5/sets';
INCLUDE '~beta/containers/v1.5/hashTable'
```

Note that since the container fragments has ORIGIN in betaenv, the above is equivalent to:

```

ORIGIN '~beta/containers/v1.5/sets'
--- program: descriptor ---
(# ...
  recordBag: @multiSet(# element:: record #);
  ...
do ...
  ...
#)

```

Examples of Use of the Container Libraries

The following will illustrate the usage of some of the container fragments. Following each example, the output from the example program is given.

Set Example

setsDemo.bet

```

ORIGIN '~beta/containers/v1.5/sets';
--- program:descriptor ---
(* This demo program illustrates the usage of the set pattern. The first part
 * of the demo illustrates inserting elements from the sets, and
 * the last part illustrates the union and symDiff operations.
 *
 * At the end of this file, a copy of the output of this program is given
 *)
(# intSet: set
  (* intSet is a set containing integerObjects *)
  (# element::< integerObject #);

  intSet1, intSet2, intSet3: @intSet;
  i: [10]^integerObject;

do (* initializing the integerObjects *)
  (for int:10 repeat &integerObject[]->i[int][]; int->i[int] for);

  (* initializing intSet1 and inserting integerObjects into it *)
  intSet1.init;
  'intSet1.capacity: '->puttext; intSet1.capacity->putInt; newline;
  (for int:4 repeat '*'->put;
    i[int][]->intSet1.insert
  for);
  (* printing the size of intSet1 *) '+'->put;
  'intSet1.size: '->puttext; intSet1.size->putInt; newline;
  (* printing intSet1 *)
  'intSet1.elements: '->puttext;
  intSet1.scan(# do current->putint; ', '->put #); newline;

  (* initializing intSet2 and inserting integerObjects into it *)
  intSet2.init;
  (for int:4 repeat
    i[int+3][]->intSet2.insert
  for);
  (* printing intSet2 *)
  'intSet2.elements: '->puttext;
  intSet2.scan(# do current->putint; ', '->put #); newline;

  (* illustrating the use of the union, diff, sect and symDiff operations *)
  'intSet2->intSet3; intSet1->intSet3.union: '->puttext;
  intSet2->intSet3; intSet1->intSet3.union;
  intSet3.scan(# do current->putint; ', '->put #); newline;

  'intSet2->intSet3; intSet1->intSet3.symDiff: '->puttext;
  intSet2->intSet3; intSet1->intSet3.symDiff;
  intSet3.scan(# do current->putint; ', '->put #); newline;

  (* illustrating finding some integerObject in insSets *)
  'intSet3.find(# predicate::< (# "elm=7" #) #): '->puttext;
  intSet3.find(# predicate::< (# do (current=7)->value #)
    do current->putint
    #); newline;
  '7->intSet3.has: '->puttext;
  (if i[7][]->intSet3.has
    //true then 'yes'->putline
    else 'no'->putline
  if);
  '10->intSet3.has: '->puttext;
  (if i[10][]->intSet3.has
    //true then 'yes'->putline

```

```

else 'no'->putline
if);

(***** OUTPUT *****)
* intSet1.capacity: -1
* intSet1.size: 4
* intSet1.elements: 4,3,2,1,
* intSet2.elements: 7,6,5,4,
* intSet2->intSet3; intSet1->intSet3.union: 1,2,3,7,6,5,4,
* intSet2->intSet3; intSet1->intSet3.symDiff: 1,2,3,7,6,5,
* intSet3.find(# predicate::< (# "elm=7" #) #): 7
* 7->intSet3.has: yes
* 10->intSet3.has: no
*****
#)

```

HashTable Example

hashTableDemo.
bet

```

ORIGIN '~beta/containers/v1.5/hashTable';
-- program: Descriptor --
(* This demo program illustrates the usage of the hashTable and
* extensibleHashTable patterns. The first part of the demo
* illustrates inserting and deleting elements from the hashtables,
* the middle part illustrates the diff and sect operations, and the
* last part of this program illustrates the possibilities for
* specifying extend the hashTable dynamically (illustrated by the
* hashTable4 object). The use of the statistics operations are also
* illustrated here.
*
* At the end of this file, a copy of the output of this program is
* given
*)
(#
  intHashTable: hashTable (* a hashTable containing integerObjects *)
    (# element::< integerObject; hashFunction::< (# do e->value #);
    #);
  intEHashTable: extensibleHashTable
    (* a hashTable containing integerObjects *)
    (# element::< integerObject; hashFunction::< (# do e->value #);
    #);
  hashtable1: @intHashTable
    (# rangeInitial::< (# do 2->value #) #);
  hashtable2: @intHashTable
    (# rangeInitial::< (# do 4->value #) #);
  hashtable3: @intHashTable;
  hashtable4: @intEHashTable
    (# rangeInitial::< (# do 2->value #) #);
  i: [10] ^integerObject;

do (* initializing the integerObjects *)
  (for int: 10 repeat
    &integerObject[]->i[int][]; int->i[int]
  for);
  (* initializing hashTable1 and putting some integerObjects into it.
  hashtable1.init;
  'hashtable1.capacity: '->puttext;
  hashtable1.capacity->putInt;
  newline;
  'hashtable1.range: '->puttext;
  hashtable1.range->putInt;
  newline;
  (for int: 4 repeat i[int][]->hashtable1.insert for);

```



```

'hashtable1.elements: '->puttext;
hashtable1.scan
  (# do current->putint; ', '->put #);
newline;
'hashtable1.size: '->puttext;
hashtable1.size->putInt;
newline;
newline;
(* initializing hashTable2 and putting some integerObjects into it. *)
hashtable2.init;
'hashtable2.capacity: '->puttext;
hashtable2.capacity->putInt;
newline;
'hashtable2.range: '->puttext;
hashtable2.range->putInt;
newline;
(for int: 4 repeat i[int+3][]->hashtable2.insert for);
'hashtable2.elements: '->puttext;
hashtable2.scan
  (# do current->putint; ', '->put #);
newline;
'hashtable2.size: '->puttext;
hashtable2.size->putInt;
newline;
newline;
(* initializing hashTable3 and putting some integerObjects into it. *)
hashtable3.init;
(for int: 4 repeat i[int+6][]->hashtable3.insert for);
'hashtable3.elements: '->puttext;
hashtable3.scan
  (# do current->putint; ', '->put #);
newline;
newline;
(* deleting integerObject 3 from hashTable3 *)
'3->hashtable3.delete: '->puttext;
i[3][]->hashtable3.delete;
hashtable3.scan
  (# do current->putint; ', '->put #);
newline;
newline;
(* illustrating the diff, and sect operations on hashTables *)
'hashtable2->hashtable3; hashtable1[]->hashtable3.diff: '->puttext;
hashtable2->hashtable3;
hashtable1[]->hashtable3.diff;
hashtable3.scan
  (# do current->putint; ', '->put #);
newline;
'hashtable2->hashtable3; hashtable1[]->hashtable3.sect: '->puttext;
hashtable2->hashtable3;
hashtable1[]->hashtable3.sect;
hashtable3.scan
  (# do current->putint; ', '->put #);
newline;
(* Illustrating the use of extensibleHashTable:
 * Initializing hashTable4 and putting some integerObjects into it.
 *)
hashtable4.init;
'hashtable4.capacity: '->puttext;
hashtable4.capacity->putInt;
newline;
'hashtable4.range: '->puttext;
hashtable4.range->putInt;
newline;

```

```

(for int: 4 repeat i[int+3][[]->hashtable4.insert for);
'hashtable4.elements: '->puttext;
hashtable4.scan
  (# do current->putint; ', '->put #);
newline;
'hashtable4.size: '->putText;
hashtable4.size->putInt;
newline;
(* illustrating extending an expensibleHashTable *)
'hashtable4.statistics(# do screen[]->print #): '->putline;
hashtable4.statistics
  (# do screen[]->print #);
(for int: hashtable4.range repeat
  int->putint;
  '->hashtable4.scanIndexed(# ... #): '->puttext;
  i[int][[]->hashtable4.hashFunction
    ->hashtable4.scanIndexed (# do current->putint; ' '->put #);
  'found'->putline;

for);
'3->hashtable4.extend(# do hashtable4.rehash #) '->putline;
3->hashtable4.extend (# do hashtable4.rehash #);
(for int: 10 repeat
  int->putint;
  '->hashtable4.has: '->puttext;
  (if i[int][[]->hashtable4.has
    // true then 'yes'->putline
    else
      'no'->putline
  if);

for);
(for int: hashtable4.range repeat
  int->putint;
  '->hashtable4.scanIndexed(# ... #): '->puttext;
  i[int][[]->hashtable4.hashFunction
    ->hashtable4.scanIndexed
      (# end::< (# do 'found'->putline #)
        do current->putint; ' '->put
        #);

for);
'hashtable4.range: '->puttext;
hashtable4.range->putInt;
newline;
'hashtable4.elements: '->puttext;
hashtable4.scan
  (# do current->putint; ', '->put #);
newline;
'hashtable4.size: '->putText;
hashtable4.size->putInt;
newline;
'hashtable4.statistics(# do screen[]->print #): '->putline;
hashtable4.statistics
  (# do screen[]->print #);
(***** OUTPUT *****)
* hashtable1.capacity: -1
* hashtable1.range: 2
* hashtable1.elements: 3,1,4,2,
* hashtable1.size: 4
*
* hashtable2.capacity: -1
* hashtable2.range: 4

```

```

* hashtable2.elements: 5,6,7,4,
* hashtable2.size: 4
*
* hashtable3.elements: 7,8,9,10,
*
* 3->hashtable3.delete: 7,8,9,10,
*
* hashtable2->hashtable3; hashtable1[]->hashtable3.diff: 5,6,7,
* hashtable2->hashtable3; hashtable1[]->hashtable3.sect: 4,
* hashtable4.capacity: -1
* hashtable4.range: 2
* hashtable4.elements: 7,5,6,4,
* hashtable4.size: 4
* hashtable4.statistics(# do screen[]->print #):
* Histogram: (2,2)
* Maximum Collisions: 2
* Minimum Collisions: 2
* Average Collisions: 2
* 1->hashtable4.scanIndexed(# ... #): 7 5 found
* 2->hashtable4.scanIndexed(# ... #): 6 4 found
* 3->hashtable4.extend(# do hashtable4.rehash #)
* 1->hashtable4.has: no
* 2->hashtable4.has: no
* 3->hashtable4.has: no
* 4->hashtable4.has: yes
* 5->hashtable4.has: yes
* 6->hashtable4.has: yes
* 7->hashtable4.has: yes
* 8->hashtable4.has: no
* 9->hashtable4.has: no
* 10->hashtable4.has: no
* 1->hashtable4.scanIndexed(# ... #): 6 found
* 2->hashtable4.scanIndexed(# ... #): 7 found
* 3->hashtable4.scanIndexed(# ... #): found
* 4->hashtable4.scanIndexed(# ... #): 4 found
* 5->hashtable4.scanIndexed(# ... #): 5 found
* hashtable4.range: 5
* hashtable4.elements: 6,7,4,5,
* hashtable4.size: 4
* hashtable4.statistics(# do screen[]->print #):
* Histogram: (1,1,0,1,1)
* Maximum Collisions: 1
* Minimum Collisions: 0
* Average Collisions: 1
*****
#)

```

List Example

```
ORIGIN '~beta/containers/v1.5/list';
```

listDemo.bet

```
--- program:descriptor ---
```

```
(* This demo program illustrates the usage of the list pattern. The first
* part of the demo illustrates inserting and deleting elements from the lists.
```

```
*
```

```
* At the end of this file, a copy of the output of this program is given
```

```
*
```

```
(# intList: list(# element::< integerObject #);
```

```
intList1, intList2: @intList; intList3: ^intList;
```

```
i: [10]^integerObject;
```

```
listPosition: ^intList3.theCellType;
```

```
do (* initializing the integerObjects *)
```

```

(for int:10 repeat &integerObject[]->i[int][][]; int->i[int] for);

(* initializing intList1 and putting some integerObjects into it *)
intList1.init;
'intList1.prepend: '->puttext;
(for int:4 repeat
  i[int][][]->intList1.prepend
for);
intList1.scan(# do current->putint; ', '->put #); newline;

(* initializing intList3 and putting some integerObjects into it *)
intList2.init;
'intList2.append: '->puttext;
(for int:4 repeat
  i[int+3][][]->intList2.append
for);
intList2.scan(# do current->putint; ', '->put #); newline;

(* illustrates the reverse scanning *)
'intList2.scanReverse: '->puttext;
intList2.scanReverse(# do current->putint; ', '->put #); newline;

(* initializing intList3 *)
&intList[]->intList3[]; intList3.init;

(* illustrating concatenation of lists *)
'intList1[]->intList2.concatenate->intList3[]: '->puttext;
intList1[]->intList2.concatenate->intList3[];
intList3.scan(# do current->putint; ', '->put #); newline;

(* illustration getting the position of some integerObject in the list
 * replacing the element at that position
 *)
'10"->("2"->intList3.at).elm[]: '->puttext;
i[2][][]->intList3.at->listPosition[];
i[10][][]->listPosition.elm[];
intList3.scan(# do current->putint; ', '->put #); newline;

(* illustrates deleting an element from the list *)
'6->intList2.at->intList2.delete: '->puttext;
i[6][][]->intList2.at->intList2.delete;
intList2.scan(# do current->putint; ', '->put #); newline;

  i[7][][]->intList3.at->listPosition[];
(* illustrating inserting integerObjects before some list position *)
'(8,7)->intList3.insertBefore: '->puttext;
(i[8][],listPosition[])->intList3.insertBefore;
intList3.scan(# do current->putint; ', '->put #); newline;

(* illustrating deleting integerObjects after some list position *)
'7->intList3.deleteAfter: '->puttext;
listPosition[]->intList3.deleteAfter;
intList3.scan(# do current->putint; ', '->put #); newline;

(***** OUTPUT *****)
* intList1.prepend: 4,3,2,1,
* intList2.append: 4,5,6,7,
* intList2.scanReverse: 7,6,5,4,
* intList1[]->intList2.concatenate->intList3[]: 4,5,6,7,4,3,2,1,
* "10"->("2"->intList3.at).elm[]: 4,5,6,7,4,3,10,1,
* 6->intList2.at->intList2.delete: 4,5,7,
* (8,7)->intList3.insertBefore: 4,5,6,8,7,4,3,10,1,
* 7->intList3.deleteAfter: 4,5,6,8,7,3,10,1,

```

```
***** )  
# )
```


Interface Description for the Container Libraries

Container.bet

```
ORIGIN '~beta/basiclib/v1.5/betaenv';
BODY 'private/containerBody';
(*
 * COPYRIGHT
 *
 *      Copyright (C) Mjolner Informatics, 1992-95
 *
 *      All rights reserved.
 *)
-- lib: Attributes --
container:
(* The top most pattern in the data structures hierarchy.
 * Container is an abstract superpattern and the currently available
 * subpatterns are (indentation specifies specialization):
 *
 *   container
 *     collection
 *       multiSet
 *       set
 *       classificationSet --- allows for sets of sets, etc.
 *       hashTable
 *       extensibleHashTable --- allows for extending the index
 *                               range
 *   list
 *     recList --- allows for recursive lists
 *   arrayContainer --- includes bubble/shell/quicksort
 *   sequentialContainer
 *     stack
 *     queue
 *     prioqueue
 *     deque
 *)
(#
  <<SLOT containerLib:Attributes>>;
  element:< object
  (* The qualification of the objects, contained in this
  * container. When using one of the subpatterns of container,
  * further binding of element is used to restrict the kinds of
  * objects, that can be inserted into the container. None of
  * the predefined container types specializes element.
  *) ;
```

```

init:<
(* Should be invoked before any usages of any type of
 * container
 *) (# do clear; INNER #);
clear:<
(* Removes all elements currently in the container, making it
 * empty
 *) (# do INNER #);
empty:< booleanValue (* Returns true if the container is empty *)
  (# do (size = 0)->value; INNER #);
size:< integerValue
(* Returns the number of elements currently in the container
 *) ;
capacity:< integerValue
(* Returns the current capacity of this container. Some
 * specializations may dynamically extend its capacity. In
 * this case, capacity returns the current capacity (subject to
 * extension at some later stage). If -1 is returned, the
 * capacity of the container is infinite (limited only by
 * memory).
 *) (# do - 1->value; INNER #);
equal:< booleanValue
(* Defines the equality test, used in the various subpatterns
 * of container in the implementation of the different
 * operations. Users of container patterns must further bind
 * equal to contain the proper equality test for the specified
 * element type. Default equality test for equal references
 * (i.e. the same object)
 *)
  (# left,right: ^element
  enter (left[],right[])
  do (left[] = right[])->value; INNER
  #);
has:< booleanValue
(* Takes an element, and checks whether it is in the container
 *)
  (#
    elm: ^element;
    doneInInner: @boolean;
    containerHasPrivate: @...
  enter elm[]
  do ('OLM:has1'->putline;*)
    INNER ;
    ('OLM:has1X'->putline;*)
    (if not doneInInner then
      ('OLM:doneInner'->putline;*)
      containerHasprivate
    if)
  #);
scan:
(* Scans through the container, invoking INNER for each
 * element in the container, for which "where" returns true.
 * "Start" is invoked at the start of the scanning, and "end"
 * is invoked at the end of the scan. In each turn of the
 * scan, "current" refers to the current element in the
 * container. The elements will be scanned in unpredictable
 * order, unless otherwise explicitly mentioned otherwise in
 * the subpatterns
 *)
  (#
    where:< elementPredicate;
    current: ^element;
    start:< object;

```



```

        end:< object;
        containerScanPrivate: @...
    ...
    #);
find:
(* Searches through the container, executing INNER for the
 * element found and returning the first element found that
 * satisfies the predicate. "Start" is invoked at the start of
 * the search, and "end" is invoked at the end of the search.
 * In each turn of the scan, "current" refers to the current
 * element in the container. If no element satisfying the
 * predicate is found, the notification "notFound" is invoked.
 *)
    (#
        predicate:< elementPredicate;
        current: ^element;
        notFound:< Notification
            (#
                do 'Element not found in container'->msg.putLine; INNER
            #);
        start:< object;
        end:< object;
        containerFindPrivate: @...
    ...
    exit current[]
    #);
copy:
(* Default copy is one-level (shallow) copying. I.e. copying
 * the container and all objects in the container. Only
 * elements satisfying "predicate" will be copied. During the
 * copying, "current" will refer to the element being
 * considered for copying. If another copying is needed, this
 * can be done by further binding "copy", and assigning "true"
 * to "doneInInner", which will result in the default copying
 * being ignored
 *)
    (#
        doneInInner: @boolean;
        theCopy: ^container;
        predicate:< elementPredicate;
        current: ^element;
        containerCopyPrivate: @...
    do INNER ;
        (if not doneInInner then
            containerCopyPrivate
        if)
    exit theCopy[]
    #);
emptyContainer: Exception
(* Invoked if some operation not valid for empty containers
 * are invoked on an empty container
 *) (# do 'Empty container'->msg.putLine; INNER #);
emptyContainerError:< emptyContainer
(* Used to handle emptyContainer errors, not handled by local
 * exception handlers in operations for containers
 *) ;
illegalCellReference:< Exception
(* Invoked if some operation tries to reference a non-existing
 * cell
 *)
    (#
        do 'Reference to nonexisting cellObject in container'-
>msg.putLine;

```

```

    INNER
    #);
elementPredicate: booleanValue
(* This pattern is used as the superpattern for all predicates
 * in the find, scan and copy operations.
 *) (# current: ^element enter current[] do true->value; INNER #);
theCellType:< (* Private *)
(* This pattern is used as the common superpattern for all
 * data structure cells, used by implementations of the
 * subpatterns of container.
 *)
  (#
    copy:<
      (# theCellCopy: ^theCellType
        do &theCellType[]->theCellCopy[]; INNER
        exit theCellCopy[]
      #)
    do INNER
  #);
implementationScanner:< (* Private *) containerImplementationScanner
implementationCopier:< (* Private *)
  (#
    theCategory:< (* Private *) container;
    thePredicate: (* Private *) ^elementPredicate;
    theCopy: (* Private *) ^theCategory
    enter (theCopy[],thePredicate[])
    ...
    exit theCopy[]
  #);
doEnter: @
  (# theOther: ^container;
    enter theOther[]
    ...
  #)
enter doEnter
(* will copy the contents of theOther[] into THIS(container)
 *)
do INNER
exit THIS(container)[]
#)
(* container *)
;
containerImplementationScanner: (* Private *)
  (#
currentElm: (* Private *) ^object;
currentCell: (* Private *) ^object;
start: (* Private *) @startOP;
startOP:< (* Private *) (# do INNER exit currentElm[] #);
next: (* Private *) @nextOP;
nextOP:< (* Private *) (# do INNER exit currentElm[] #);
more: (* Private *) @moreOP;
moreOP:< (* Private *) booleanValue
do INNER
#)

```

Collection.bet

```
ORIGIN 'container';
```

```

BODY 'private/collectionBody';
(*
 * COPYRIGHT
 *
 *      Copyright (C) Mjolner Informatics, 1992-94
 *
 *      All rights reserved.
 *)
--- lib: attributes ---
collection: container
(* Collection is the superpattern for all collection data
 * structures. Defines the operations:
 *   insert, delete, union, diff, sect, xclOr
 * Container is an abstract superpattern and the currently available
 * subpatterns are (indentation specifies specialization):
 *)
 *      collection
 *      multiSet
 *      set
 *      classificationSet --- allows for sets of sets, etc.
 *      hashTable
 *      extensibleHashTable --- allows for extending the index
 *                               range
 *)
(# <<SLOT collectionLib: attributes>>;
insert:<
(* Takes an element and inserts it in THIS(collection) *)
(# elm: ^element
enter elm[]
do INNER
#);
delete:<
(* Takes an element and removes the object from
 * THIS(collection). Invokes the notification notFound, if the
 * element cannot be found.
 *)
(# elm: ^element;
notFound:< Notification
(#
do 'Element not found in container'->msg.putLine;
INNER
#);
enter elm[]
do INNER
#);
union:<
(* Takes a collection, and unifies its elements into
 *      THIS(collection)
 *)
(# theOther: ^collection
enter theOther[]
...
exit this(collection)[]
#);
diff:<
(* Takes a collection, and removes its elements from
 *      THIS(collection)
 *)
(# theOther: ^collection
enter theOther[]
...
exit this(collection)[]
#);

```

```

sect:<
  (* Takes a collection, and keeps in THIS(collection) those
   * elements that are in both collections
   *)
  (# theOther: ^collection
  enter theOther[]
  ...
  exit this(collection)[]
  #);
symDiff:<
  (* Takes a collection, and inserts into THIS(collection) those
   * elements not being in both collections
   *)
  (# theOther: ^collection; tmp1, tmp2: ^collection
  enter theOther[]
  ...
  exit this(collection)[]
  #);
theCellType::< (* Private *)
  (# elm: ^element;
   copy::< (# do elm[]->theCellCopy.elm[]; INNER #);
  enter elm[]
  do INNER
  exit elm[]
  #);
implementationCopier::< (* Private*)
  (# theCategory::< collection do INNER #)
do INNER
#) (* collection *);

```

Sets.bet

```

ORIGIN 'collection';
BODY 'private/setsBody';
(*
 * COPYRIGHT
 * Copyright (C) Mjolner Informatics, 1992-94
 * All rights reserved.
 *)
--- lib: attributes ---
multiSet: collection
  (* MultiSet is a data structure for collecting elements (duplicates
   * are allowed of the same element). Defines no new operations.
   * MultiSet is the superpattern for Set, which ignores duplicates.
   * The available subpatterns are (indentation specifies
   * specialization):
   * multiSet
   * set
   * classification
   *)
  (# <<SLOT multiSetLib: attributes>>;
  clear::< (# ... #);
  size::< (# ... #);
  insert::<
    (* inserting in a multiSet is somewhat dependent on the
     * particular subpattern. In some cases, insertion in only
     * done if the element is not already in the multiSet
     * (e.g. sets). The local attribute, allowedToInsert, controls

```

```

    * whether the element is actually inserted in the multiSet.
    *)
    (# allowedToInsert: @boolean;
     multiSetInsertPrivate: @...
    do true->allowedToInsert;
     (*'OLM: starting insert' ->putline;*)
     INNER;
     (*'OLM: 1. INNER after'->putline;*)
     (if allowedToInsert then multiSetInsertPrivate if)
    #);
delete::< (# multiSetDeletePrivate: @... do multiSetDeletePrivate; INNER #);
theCellType::< (* Private *)
    (# pred, succ: (* Private *) ^theCellType;
     remove: (* Private *) (# ... #)
    do INNER #);
implementationScanner::< (* Private *)
    (# multiSetImplementationScannerPrivate: @...;
     startOP::< (* Private *)
       (# theCell: ^theCellType
        ...
       #);
     nextOP::< (* Private *)
       (# theCell: ^theCellType
        ...
       #);
     moreOP::< (* Private *)
       (# ... #)
    #);
implementationCopier::< (* Private *)
    (# theCategory::< multiSet;
     doneInInner: @boolean;
     multiSetImplementationCopierPrivate: @...
    do INNER;
     (if not doneInInner then multiSetImplementationCopierPrivate
if)
    #);
storage: (* Private *) @...
do INNER
#) (* multiSet *);

(*-----*)
(*--- set: multiSet -----*)
(*-----*)

set: multiSet
(* A set is a multiSet in which duplicates are removed. Defines no
 * new operations
 *)
(# <<SLOT setLib: attributes>>;
 insert::<
  (* Inserts only the element if it is not in the set already *)
  (# ... #);
 implementationCopier::< (* Private *)
  (# theCategory::< set do INNER #);
do INNER
#) (* set *);

```

Classification.set

```

ORIGIN 'sets';
BODY 'private/classificationBody';
(*
 * COPYRIGHT
 *     Copyright (C) Mjolner Informatics, 1992-94
 *     All rights reserved.
 *)
--- lib: attributes ---
classificationSet: set
(* A classification set is a set in which subsets may be defined.
 * The subsets of a classificationSet are such that inserting
 * elements in the subsets, makes the element a member of the
 * classificationSet. Inserting an element in one of the subsets of
 * the classificationSet, which is already in the classificationSet
 * (or one of its subsets), will move the element from the
 * classificationSet (or the subset) into the proper subset. If a
 * classificationSet is inserted as a subset of
 * THIS(classificationSet), it is ensured that inserting elements in
 * the subset will not result in duplicates being present in
 * THIS(classificationSet). The reference SuperSet in
 * THIS(classificationSet) refers to the classificationSet,
 * THIS(classificationSet) might be a subset of. SuperSet is NONE,
 * if THIS(classificationSet) is not a subset of any other
 * classificationSet. Defines two new operations:
 *     insertSubset, scanUnclassified
 *)
(# <<SLOT classificationSetLib: attributes>>;
clear::< (# ... #);
size::< (# ... #);
subsets: @
(* The subsets currently registered in THIS(classificationSet)
 *)
set(# element:< classificationSet #);
superSet: ^classificationSet;
insertSubset:
(* Takes a classificationSet, and defines it as a subset of
 * this classificationSet. The elements of this subset will
 * now be member of THIS(classificationSet)
 *)
(# ss: ^classificationSet;
  illegalSubset:< Exception
    (* Will be invoked if a subset is inserted with an
    * element type is not a subpattern of the element type of
    * THIS(classificationSet).
    *)
  (#
    do 'The elements of the subset does not share '->msg.putText;
      'qualifications with the elements of this set'-
>msg.putLine;
    INNER
      #)
    enter ss[]
    ...
  #);
insert::< (# ... #);
delete::< (# ... #);
scanUnclassified:
(* similar to scan, except that it only scans those elements
 * that are NOT member of any subsets

```

```

*)
(# where:< elementPredicate;
  current: ^element;
  start:< object;
  end:< object
do start; ...; end
#);
implementationScanner::< (* Private *)
  (# currentSubset: (* Private *) ^subsets.theCellType;
  theSubsetScanner: (* Private *) ^containerImplementationScanner;
  startOP::< (* Private *)
    (# ... #);
  nextOP::< (* Private *)
    (# ... #)
#);
implementationCopier::< (* Private *)
  (# theCategory::< classificationSet
  ...
#)
#) (* classificationSet *);

```

ArrayContainer.bet

```

ORIGIN 'container';
BODY 'private/arrayContainerBody';
(*
* COPYRIGHT
* Copyright (C) Mjolner Informatics, 1992-94
* All rights reserved.
*)
--- lib: attributes ---
arrayContainer: container
(* ArrayContainer is an abstraction of a repetition, offering
* container capabilities as well as repetition capabilities.
* Furthermore, arrayContainer implements three popular sorting
* algorithms on the elements in the arrayContainer. Defines the
* operations:
* capacityInitial, capacityIncrement, capacityExtend,
* get, put, delete, bubbleSort, shellSort, quickSort
*)
(# <<SLOT arrayContainerLib: attributes>>;
less:< booleanValue
  (* should be further bound to contain an ordering operation to
  * be used by the sorting algorithms. Less is only used during
  * sorting, and can therefore be ignored if sorting is not
  * applied.
  *)
  (# left, right: ^element
  enter (left[],right[])
  do INNER
  #);
clear::< (# ... #);
size::< (# ... #);
capacity::< (# ... #);
capacityInitial:< integerObject(# do 25->value; INNER #);
capacityIncrement:< integerObject(# do 10->value; INNER #);
capacityExtend:< (# ... #);
put:<

```

```

    (* Takes an element and an index position and inserts the
    * element at that position
    *)
    (# elm: ^element; inx: @integer
    enter (elm[],inx)
    ...
    #);
get:<
    (* Takes an index position and returns the element at that
    * position
    *)
    (# elm: ^element; inx: @integer
    enter inx
    ...
    exit elm[]
    #);
delete:<
    (* Takes an index position and deletes the element at that
    * position (the deleted element is returned)
    *)
    (# inx: @integer; elm: ^element;
    enter inx
    ...
    exit elm[]
    #);
bubbleSort:
    (* Takes an index position and sorts the elements in positions
    * [1: n], using the bubblesort algorithm
    *)
    (# n: @integer
    enter n
    do ...; INNER
    exit this(arrayContainer)[]
    #);
shellSort:
    (* Takes an index position and sorts the elements in positions
    * [1: n], using the shellsort algorithm
    *)
    (# n: @integer
    enter n
    do ...; INNER
    exit this(arrayContainer)[]
    #);
quickSort:
    (* Takes an index position and sorts the elements in positions
    * [1: n], using the quicksort algorithm
    *)
    (# n: @integer;
    enter n
    do ...; INNER
    exit this(arrayContainer)[]
    #);
theCellType::< (* Private *)
    (# elm: ^element; occupied: @boolean;
    copy::< (#
    do elm[]->theCellCopy.elm[];
    occupied->theCellCopy.occupied; INNER
    #)
    enter elm[]
    do INNER
    exit elm[]
    #);
implementationScanner::< (* Private *)

```



```

    (# i: (* Private *) @integer;
     startOP::< (* Private *)
       (# theCell: ^theCellType
        ...
        #);
     nextOP::< (* Private *)
       (# theCell: ^theCellType
        ...
        #);
     moreOP::< (* Private *)
       (# ...
        #)
    #);
implementationCopier::< (* Private *)
  (# theCategory::< arrayContainer; current: ^theCellType
   ... #);
storage: (* Private *) @...
do INNER
#) (* arrayContainer *);

```

HashTable.bet

```

ORIGIN 'collection';
BODY 'private/hashTableBody';
(*
 * COPYRIGHT
 *      Copyright (C) Mjolner Informatics, 1992-94
 *      All rights reserved.
 *)
--- lib:attributes ---
hashTable: collection
(* A hashTable is an efficient data structure for storing a unknown
 * number of elements for quick access. Defines the following new
 * operations: range, hashFunction, scanIndexed, findIndexed and
 * statistics. The available subpatterns are (indentation specifies
 * specialization):
 *      hashTable
 *      extensibleHashTable
 *)
(# <<SLOT hashTableLib:attributes>>;
clear::< (# ... #);
size::< (# ... #);
rangeInitial:< integerValue
  (* This pattern is used to define the value domain for the
   * hashFunction and thus the number of indices in
   * this(hashTable).
   *)
  (# do 25->value; INNER #);
range: integerValue
  (* This pattern returns the number of indices in
   * this(hashTable).
   *)
  (# ... #);
indexValue: integerValue
  (#
  do INNER;
  (if value < 0 then -value->value if);
  value mod range->value;

```

```

        (if value = 0 then range->value if)
    #);
index: indexValue(# enter value do INNER #); theIndex: @index;
hashFunction:< indexValue
    (* This pattern is invoked on all elements inserted into
     * THIS(hashTable). The default behaviour will generate many
     * collisions, and specialization is therefore strongly advised
     *)
    (# e: ^element
     enter e[]
     do 0->value;
       INNER;
     #);
has::< (# ... #);
insert::< (# ... #);
delete::< (# ... #);
scanIndexed:
    (* enters an index value, and scans all elements in
     * this(hashTable) with the same index value (i.e. all
     * collisions on that index value. That is, scanIndexed is
     * similar to scan, except that it only scans the elements
     * which happen to be indexed at the same index in the
     * hashTable. Each time INNER is invoked, current refers to
     * the actual element. Note that if you want to scan all
     * element with the same index as some known element, elm, you
     * can do so by: elm[]->table.hashFunction->table.scanIndexed(#
     * ... do ... #)
     *)
    (# inx: @integer;
     where:< elementPredicate;
     current: ^element;
     start:< object;
     end:< object;
     hashtableScanIndexedPrivate: @...
     enter theIndex->inx
     ...
     #);
findIndexed:
    (* enters an index value, and seeks among the elements for the
     * first element, for which predicate holds true. If such an
     * element is found, INNER is called, and then that element is
     * returned by findIndexed. While INNER is invoked, current
     * refers to the element found. Note that findIndexed behaves
     * somewhat similar to Find, except that findIndexed only scans
     * through the elements in the hashTable with the given hash
     * index (i.e. the collisions). This implies that findIndexed
     * takes advantage of the fact that this is a hashTable,
     * meaning that findIndexed is a lot faster than Find. Note
     * that if you want to find some element with the same index as
     * some known element, elm1, you can do so by:
     * elm1[]->table.hashFunction ->table.findIndexed(#
     * predicate:< (# ... do ... #) #); ->elm2[]
     *)
    (# inx: @integer;
     predicate:< elementPredicate;
     current: ^element;
     notFound:< Notification
     (#
       do 'Element not found in hashtable'->msg.putline;
       INNER
     #);
     start:< object;
     end:< object;

```

```

        hashtableFindIndexedPrivate: @...
enter theIndex->inx
...
exit current[]
#);
statistics:
(* calculates statistics on the current status of
 * this(hashTable). Returns a histogram in the form of a
 * table, where each entry in the table contains the number on
 * collisions for that hash index. Also returned are the
 * maximum, minimum and average number of collisions found in
 * the hashTable. The local print pattern prints this
 * information on some stream
 *)
(# histogram: [range] @integer;
 max, min, average, usedIndices: @integer;
 print:
  (# s: ^stream; i: @integer
   enter s[]
   do 'Histogram: '->s.putText;
     ('->s.put;
     l->i; histogram[i]->s.putInt;
     loop: (if i<range then
            ', '->s.put;
            i+1->i; histogram[i]->s.putInt;
            restart loop
          if);
     ')'->s.putline;
     'Maximum Collisions: '->s.putText;max->s.putInt;s.newline;
     'Minimum Collisions: '->s.putText;min->s.putInt;s.newline;
     'Average Collisions: '->s.putText;average->s.putInt;s.newline;
     INNER
   #)
do maxInt->min; minInt->max;
 (for i:range repeat
  i->scanIndexed(# do histogram[i]+1->histogram[i] #)
 for);
 (for i:range repeat
  (if histogram[i]>max then histogram[i]->max if);
  (if histogram[i]<min then histogram[i]->min if);
  (if histogram[i]>0 then
   average + histogram[i]->average;
   usedIndices+1->usedIndices
  if);
 for);
 (if usedIndices = 0 then 0->average
  else average div usedIndices->average
 if);
 INNER
 exit (histogram, max, min, average, usedIndices)
 #);
theCellType::: (< (* Private *)
 (# next: (* Private *) ^theCellType #));
implementationScanner::: (< (* Private *)
 (# i: @integer;
  startOP::: (< (* Private *)
   (# theCell: ^theCellType
   ...
   #);
  nextOP::: (< (* Private *)
   (# theCell: ^theCellType
   ...
   #);

```

```

        moreOP::< (* Private *)
            (# ... #)
    #);
implementationCopier::< (* Private *)
    (# theCategory::< hashTable
        ...
    #);
storage: (* Private *) @...
do INNER
#) (* hashTable *);

(*-----*)
(*--- extensibleHashTable-----*)
(*-----*)

ExtensibleHashTable: hashTable
(* ExtensibleHashTable makes it possible to extend the range of
 * index values dynamically.  If the range of index value is
 * extended, the user of this(extensibleHashTable) needs to take
 * special care that the hash index of the existing objects in the
 * table is not changed, otherwise the table may not work properly.
 * In order to cope with hashFunctions that might be dependent on
 * the range of index values, extensibleHashTable defines a rehash
 * function which may be invoked to rearrange the hash indeces of
 * the existing elements in the table.  Note that rehash'ing the
 * table is a relatively costly operation, and should therefore only
 * be invoked when absolutely needed.  To make extending a table
 * safe from eventual changes in the hash indeces of elements
 * already in the tabel, the following is advised: table.extend(# do
 * table.rehash #) Defines two new operations: extend, rehash
 *)
(# <<SLOT extensibleHashTableLib:attributes>>;
extend:
    (* extends this(extensibleHashTable) with the given number of
     * index values
     *)
    (# increment: @integer
     enter increment
     ...
    #);
rehash:
    (* takes all elements in this(extensibleHashTable) and
     * calculates new hash indexes for all of them
     *)
    (# ... #);
implementationCopier::< (* Private *)
    (# theCategory::< extensibleHashTable do INNER #)
do INNER
#) (* extensibleHashTable *);

```

SeqContainers.beta

```

ORIGIN 'container';
BODY 'private/seqContainersBody';
(*
 * COPYRIGHT
 *     Copyright (C) Mjolner Informatics, 1992-94
 *     All rights reserved.

```

```

*)
--- lib: attributes ---
sequentialContainer: container
  (* sequentialContainer is an abstract superpattern and the
  * currently available subpatterns are (indentation specifies
  * specialization):
  *   sequentialContainer
  *   stack
  *   queue
  *   prioqueue
  *   deque
  *)
  (# <<SLOT sequentialContainerLib: attributes>>;
  clear::< (# ... #);
  size::< (# ... #);
  theCellType::< (* Private *)
    (# succ, pred: (* Private *) ^theCellType;
    elm: ^element;
    copy::< (# do elm[]->theCellCopy.elm[]; INNER #);
    do INNER #);
  implementationScanner::< (* Private *)
    (# sequentialContainerImplementationScannerPrivate: @...;
    startOP::< (# theCell: ^theCellType ... #);
    nextOP::< (# theCell: ^theCellType ... #);
    moreOP::< (* Private *) (# ... #)
    #);
  implementationCopier::< (* Private *)
    (# theCategory::< sequentialContainer do INNER #);
  storage: (* Private *) @...
  do INNER
  #) (* sequentialContainer *);

(*-----*)
(*--- stack-----*)
(*-----*)

stack: sequentialContainer
  (* Stack is an ordinary stack data structure. Defines the
  * operations: top, push, pop
  *)
  (# <<SLOT stackLib: attributes>>;
  top: (* Returns the topmost element on THIS(stack) *)
    (# elm: ^element; empty:< emptyContainer
    ...
    exit elm[]
    #);
  push:
    (* Takes an element and places it on the top of THIS(stack) *)
    (# elm: ^element
    enter elm[]
    ...
    #);
  pop:
    (* Returns the topmost element, and removes it from the stack
    *)
    (# elm: ^element; empty:< emptyContainer
    ...
    exit elm[]
    #);
  implementationCopier::< (* Private *)
    (# theCategory::< stack ... #)
  do INNER
  #) (* stack *);

```

```

(*-----*)
(*--- queue -----*)
(*-----*)

queue: sequentialContainer
(* Queue is an ordinary queue data structure. Defines the
 * operations: front, insert, remove
 *)
(# <<SLOT queueLib: attributes>>;
 front: (* Returns the element in the front of THIS(queue) *)
  (# elm: ^element; empty:< emptyContainer
  ...
  exit elm[]
  #);
 insert:
  (* Takes an element and inserts it at the end of THIS(queue) *)
  (# elm: ^element
  enter elm[]
  ...
  #);
 remove:
  (* Returns the element in front, and removes it from the queue
  *)
  (# elm: ^element; empty:< emptyContainer
  ...
  exit elm[]
  #);
 implementationCopier::< (* Private *)
  (# theCategory::< queue ... #)
do INNER
#) (* queue *);

(*-----*)
(*--- prioQueue -----*)
(*-----*)

prioQueue: sequentialContainer
(* PrioQueue is a priority queue, in which the elements are kept in
 * separate queues for each priority. Defines the operations:
 * front, insert, remove
 *)
(# <<SLOT prioQueueLib: attributes>>;
 size::< (# ... #);
 front: (* Returns the element in front with the given priority *)
  (# prio: @integer; elm: ^element; empty:< emptyContainer
  enter prio
  ...
  exit elm[]
  #);
 insert:
  (* Insert an element in the queue with the given priority *)
  (# prio: @integer; elm: ^element
  enter (elm[],prio)
  ...
  #);
 remove:
  (* Remove and returns the front element with the given
  * priority
  *)
  (# prio: @integer; elm: ^element; empty:< emptyContainer
  enter prio
  ...

```

```

    exit elm[]
    #);
scanPriority:
    (* scans through all elements of the given priority *)
    (# prio: @integer;
     where:< elementPredicate;
     current: ^element;
     start:< object;
     end:< object;
     sequentialContainerPrioQueueScanPriorityPrivate: @...
    enter prio
    ...
    #);
theCellType:::< (* Private *)
    (# prio: @integer;
     theQueue: ^queue;
     copy::: (< (#
                do prio->theCellCopy.prio;
                (if theQueue[] = NONE then
                 else theQueue.copy->theCellCopy.theQueue[]
                if)
                #)
            do INNER #);
implementationScanner:::< (* Private *)
    (# currentQueue: ^theCellType;
     theQueueScanner: (* Private *) ^containerImplementationScanner;
     startOP::: (< (* Private *)
                (# ... #);
     nextOP::: (< (* Private *)
                (# ... #)
    #);
implementationCopier:::< (* Private *)
    (# theCategory::: prioQueue;
     theQueueCopy: ^queue;
     ...
    #)
do INNER
#) (* prioQueue *);

(*-----*)
(*--- deque -----*)
(*-----*)

deque: sequentialContainer
    (* Deque is a double-ended queue in which elements may be inserted
    * and removed from both ends of the queue. Defines the operations:
    * front, insertFront, removeFront, back, insertBack, removeBack
    *)
    (# <<SLOT dequeLib: attributes>>;
     front: (* Returns the element in the front *)
        (# elm: ^element; empty:< emptyContainer
         ...
         exit elm[]
         #);
     insertFront: (* Takes an element and inserts it at the front *)
        (# elm: ^element
         enter elm[]
         ...
         #);
     removeFront:
        (* Remove and return the element in front of THIS(deque) *)
        (# elm: ^element; empty:< emptyContainer
         ...

```

```

        exit elm[]
        #);
back: (* Returns the element in the back *)
        (# elm: ^element; empty:< emptyContainer
        ...
        exit elm[]
        #);
insertBack: (* Takes an element and inserts it at the back *)
        (# elm: ^element
        enter elm[]
        ...
        #);
removeBack: (* Returns the element in the back, removing it *)
        (# elm: ^element; empty:< emptyContainer
        ...
        exit elm[]
        #);
implementationCopier::< (* Private *)
        (# theCategory::< deque;
        ...
        #)
do INNER
#) (* deque *);

```

List.bet

```

ORIGIN 'container';
BODY 'private/listBody';
(*
 * COPYRIGHT
 * Copyright (C) Mjolner Informatics, 1992-94
 * All rights reserved.
 *)
--- lib: attributes ---
list: container
(* This list pattern defines a double-linked list data structure.
 * Most operations either enters or exits a list position. List
 * positions are references into a particular place in the list,
 * containing an element. These positions are instances of
 * theCellType.
 * list
 * recList --- allows for recursive lists
 *
 * Defines the following new operations: prepend, append, head,
 * tail, last, preamble, at, locate, concatenate, splitBefore,
 * splitAfter, insertBefore, insertAfter, delete, deleteBefore,
 * deleteAfter, scanReverse, iterate, iterateFrom,
 * iterateReverse, iterateReverseFrom
 *)
(# <<SLOT listLib: attributes>>;
theCellType::<
(* theCellType is the pattern from which the individual list
 * positions are created. It defines the succ and pred
 * references to the list positions immediately before/after
 * this position in the list (NONE means the end of the list
 * (either end). The elm attribute refers to the element at
 * this position in the list.
 *)

```



```

    (# succ, pred: ^theCelltype;
      elm: ^element;
      copy::< (# do elm[]->theCellCopy.elm[]; INNER #);
    do INNER
    #);
clear::< (# ... #);
size::< (# ... #);
prepend: (* insert elm as first element *)
  (# elm: ^element; position: ^theCellType
  enter elm[]
  ...
  exit position[]
  #);
append: (* insert elm as last element *)
  (# elm: ^element; position: ^theCellType
  enter elm[]
  ...
  exit position[]
  #);
head:
  (* returns the position element of the first position in the
  * list
  *)
  (# position: ^theCellType
  ...
  exit position[]
  #);
tail:
  (* returns a copy of THIS(list), except the first element *)
  (# lst: ^list
  ...
  exit lst[]
  #);
last:
  (* returns the position element of the last position in the
  * list
  *)
  (# position: ^theCellType
  ...
  exit position[]
  #);
preample:
  (* returns a copy of THIS(list), except the last element *)
  (# lst: ^list
  ...
  exit lst[]
  #);
concatenate:
  (* returns a new list, containing the concatenated list *)
  (# theOther, lst: ^list; listConcatenatePrivate: @...
  enter theOther[]
  do listConcatenatePrivate; INNER
  exit lst[]
  #);
insertBefore: (* if position=NONE, insert elm as last element *)
  (# elm: ^element; position, newPosition: ^theCellType
  enter (elm[], position[])
  ...
  exit newPosition[]
  #);
insertAfter: (* if position=NONE, insert elm as first element *)
  (# elm: ^element; position, newPosition: ^theCellType
  enter (elm[], position[])

```

```

...
exit newPosition[]
#);
delete: (* if position=NONE, delete nothing *)
(# deletedPosition, position: ^theCellType;
  empty:< emptyContainer
  enter position[]
  ...
  exit deletedPosition[]
#);
deleteBefore:
(* if position=NONE, delete last position element *)
(# deletedPosition, position: ^theCellType;
  empty:< emptyContainer
  enter position[]
  ...
  exit deletedPosition[]
#);
deleteAfter:
(* if position=NONE, delete first position element *)
(# deletedPosition, position: ^theCellType;
  empty:< emptyContainer
  enter position[]
  ...
  exit deletedPosition[]
#);
splitBefore:
(* splits this(list) into two lists, where the elements before
 * position (excluding position) is placed in preList and the
 * rest of this(list) is placed in postList. PositionElm will
 * be the head of postList. If position=NONE, preList will
 * become a copy of the entire list and postList will become
 * NONE
 *)
(# position: ^theCellType;
  preList, postList: ^list;
  listSplitBeforePrivate: @...
  enter position[]
  do listSplitBeforePrivate; INNER
  exit (preList[], postList[])
#);
splitAfter:
(* splits this(list) into two lists, where the elements before
 * position (including position) is placed in preList and the
 * rest of this(list) is placed in postList. PositionElm will
 * be the last element in postList If position=NONE, preList
 * will become a copy of the entire list and postList will
 * become NONE
 *)
(# position: ^theCellType;
  preList, postList: ^list;
  listSplitAfterPrivate: @...
  enter position[]
  do listSplitAfterPrivate; INNER
  exit (preList[], postList[])
#);
at: (* returns the position of elm in the list *)
(# elm: ^element;
  position: ^theCellType
  enter elm[]
  ...
  exit position[]
#);

```

```

locate:
(* returns the position of the element in the list, satisfying
 * predicate. This operation is similar to find, except that
 * it returns the position, not the element.
 *)
(# predicate:< cellPredicate;
 notFound:< Notification
  (#
    do 'Element not found in list'->msg.putLine;
    INNER
  #);
 start:< object;
 end:< object;
 position: ^theCellType;
 listLocatePrivate: @...
 ...
 exit position[]
 #);

scanReverse:
(* similar to scan, except that it scans the list in reverse
 * direction
 *)
(# where:< elementPredicate;
 current: ^element;
 start:< object;
 end:< object;
 listScanReversePrivate: @...
 ...
 #);

scanFrom:
(* similar to scan, except that it scans the list from the
 * given position
 *)
(# where:< elementPredicate;
 position: ^theCellType;
 current: ^element;
 start:< object;
 end:< object;
 listScanFromPrivate: @...
 enter position[]
 ...
 #);

scanReverseFrom:
(* similar to scanReverse, except that it scans the list in
 * reverse direction from the given position
 *)
(# where:< elementPredicate;
 position: ^theCellType;
 current: ^element;
 start:< object;
 end:< object;
 listScanReverseFromPrivate: @...
 enter position[]
 ...
 #);

iterate:
(* similar to scan, except that "current" refers to the
 * current cell in the list. If furthermore the list is an
 * instance of the recList subpattern, iterate respects the
 * sublist structure by not scanning the elements in the
 * sublists (as scan does). I.e. during an iterate of a
 * recList, "current" may refer to a cell containing either an
 * element, or a sublist.

```

```

    *)
    (# where:< cellPredicate;
      current: ^theCellType;
      start:< object;
      end:< object;
      listIteratePrivate: @...
      ...
    #);
iterateFrom:
    (* similar to iterate, except that it takes a position, and
      * starts the iteration from that position.
      *)
    (# where:< cellPredicate;
      position: ^theCellType;
      current: ^theCellType;
      start:< object;
      end:< object;
      listIterateFromPrivate: @...
    enter position[]
      ...
    #);
iterateReverse:
    (* similar to iterate, except that it iterates backwards.
      *)
    (# where:< cellPredicate;
      current: ^theCellType;
      start:< object;
      end:< object;
      listIterateReversePrivate: @...
      ...
    #);
iterateReverseFrom:
    (* similar to iterateReverse, except that it takes a position,
      * and starts the reverse iteration from that position.
      *)
    (# where:< cellPredicate;
      position: ^theCellType;
      current: ^theCellType;
      start:< object;
      end:< object;
      listIterateReverseFromPrivate: @...
    enter position[]
      ...
    #);
cellPredicate: booleanValue
    (* This pattern is used as the superpattern for the predicates
      * in the locate and iterate operations.
      *)
    (# current: ^theCellType
    enter current[]
    do true->value;
      INNER
    #);
implementationScanner::< (* Private *)
    (# listImplementationScannerPrivate: @
      ...;
      startOP::< (* Private *)
        (# theCell: ^theCellType
          ...
        #);
      nextOP::< (* Private *)
        (# theCell: ^theCellType
          ...

```

```

        #);
        moreOP::< (* Private *)
            (# ... #)
    #);
implementationCopier::< (* Private *)
    (# theCategory::< list
        ...
    #);
storage: (* Private *) @...
do INNER
#) (* list *);

```

RecList.bet

```

ORIGIN 'list';
BODY 'private/recListBody';
(*
 * COPYRIGHT
 *
 * Copyright (C) Mjolner Informatics, 1992-94
 *
 * All rights reserved.
 *)
--- lib: attributes ---
recList: list
(* This recList pattern extends the list pattern to allow for
 * sublists. Most operations either enters or exits a list
 * position. List positions in a recList may contain either an
 * element or a sublist. Defines the following new operations:
 *
 * insertSublistBefore, insertSublistAfter,
 *)
(# <<SLOT recListLib: attributes>>;
theCellType::<
    (* theCellType is the pattern from which the individual
    * recList positions are created. It also defines the when
    * attribute which is used for accessing either the element or
    * the sublist at this position. By further binding element,
    * respectively sublist, in when, the actions to be executed in
    * either case, may be specified.
    *)
    (# sublist: ^list;
    when:
        (* to differentiate between this(theCellType) holding an
        * element or a sublist
        *)
        (# elm:< (# elm: ^element
            ...
            exit elm[]
            #);
        sublist:< (# lst: ^list
            ...
            exit lst[]
            #);
        recListCellWhenPrivate: @...
        do INNER; recListCellWhenPrivate
        #);
    copy::< (#

```

```

        do (if sublist[] = NONE then
            else sublist.copy->theCellCopy.sublist[]
            if)
        #)
    do INNER
    #);
insertSublistBefore:
    (* if position=NONE, insert lst as last element *)
    (# lst: ^list; position, newPosition: ^theCellType;
     illegalSublist:< illegalSublistException
     enter (lst[], position[])
     ...
     exit newPosition[]
     #);
insertSublistAfter:
    (* if position=NONE, insert lst as first element *)
    (# lst: ^list; position, newPosition: ^theCellType;
     illegalSublist:< illegalSublistException
     enter (lst[], position[])
     ...
     exit newPosition[]
     #);
illegalSublistException: Exception
    (#
    do 'The element of this sublist does not share '->msg.putText;
      'qualifications with the element type of the list'-
    >msg.putLine;
    INNER
    #);
implementationScanner::< (* Private *)
    (# sublistScanner: (* Private *) ^containerImplementationScanner;
     sublistPosition: (* Private *) ^theCellType;
     sublistNextPosition: (* Private *) ^theCellType;
     startOP::< (* Private *)
     (# theCell: ^theCellType;
     ...
     #);
     nextOP::< (* Private *)
     (# theCell: ^theCellType;
     ...
     #)
    #);
implementationCopier::< (* Private *)
    (# theCategory::< recList do INNER #)
do INNER
#) (* recList *);

```

Index

The entries in this index are the identifiers defined in the public interface of the libraries: The minor level entries refer to identifiers defined local to the identifier of the major level entry. For those index entries referring to patterns with super- or subpatterns within the library, these patterns are specified in special sections of the minor level index for that identifier.

A

append.....	37
arrayContainer	27
. booleanValue	
. . subpatterns	
. . . less	27
. bubbleSort.....	28
. capacity.....	27
. capacityExtend	27
. capacityIncrement	27
. . superpattern	
. . . integerObject	42
. capacityInitial.....	27
. . superpattern	
. . . integerObject	42
. clear	27
. delete.....	28
. get.....	28
. implementationCopier.....	29
. implementationScanner.....	28
. integerObject	
. . subpatterns	
. . . capacityIncrement.....	27
. . . capacityInitial	27
. less.....	27
. . superpattern	
. . . booleanValue.....	42
. put	27
. quickSort.....	28
. shellSort.....	28
. size	27
. storage.....	29
. superpattern	
. . container.....	27
. theCellType	28
ArrayContainer	5
at 38	

B

back.....	36
bubbleSort.....	28

C

capacity.....	20, 27
capacityExtend	27
capacityIncrement.....	27
capacityInitial.....	27
cellPredicate	40
classificationSet.....	26
. clear.....	26
. delete.....	26
. implementationCopier	27
. implementationScanner	27
. insert.....	26
. insertSubset.....	26
. scanUnclassified	26
. size.....	26
. subsets	26
. superpattern	
. . set.....	26
. superSet.....	26
ClassificationSet	3
clear	20, 24, 26, 27, 29, 33, 37
collection.....	23
. delete.....	23
. diff	23
. implementationCopier	24
. insert.....	23
. sect.....	24
. subpatterns	
. hashCode	29
. multiSet	24
. superpattern	
. . container.....	23
. symDiff.....	24
. theCellType.....	24
. union	23
Collection.....	2
concatenate.....	37
container	19
. booleanValue	
. . subpatterns	
. . . elementPredicate.....	22
. . . empty	20
. . . equal	20

. . . has.....	20
. capacity.....	20
. . superpattern	
. . . integerValue.....	42
. clear.....	20
. copy.....	21
. doEnter.....	22
. element.....	19
. elementPredicate.....	22
. . superpattern	
. . . booleanValue.....	42
. empty.....	20
. . superpattern	
. . . booleanValue.....	42
. emptyContainer.....	21
. . superpattern	
. . . Exception.....	42
. emptyContainerError.....	21
. equal.....	20
. . superpattern	
. . . booleanValue.....	42
. Exception	
. . subpatterns	
. . . emptyContainer.....	21
. . . illegalCellReference.....	21
. find.....	21
. has.....	20
. . superpattern	
. . . booleanValue.....	42
. illegalCellReference.....	21
. . superpattern	
. . . Exception.....	42
. implementationCopier.....	22
. implementationScanner.....	22
. init.....	20
. integerValue	
. . subpatterns	
. . . capacity.....	20
. scan.....	20
. size.....	20
. subpatterns	
arrayContainer.....	27
collection.....	23
list.....	36
sequentialContainer.....	33
. theCellType.....	22
Container.....	1
container libraries.....	1
containerImplementationScanner.....	22
. currentCell.....	22
. currentElm.....	22
. more.....	22
. moreOP.....	22
. next.....	22
. nextOP.....	22
. start.....	22
. startOP.....	22
copy.....	21
currentCell.....	22
currentElm.....	22
D	
delete.....	23, 25, 26, 28, 30, 38
deleteAfter.....	38
deleteBefore.....	38
deque.....	35
. back.....	36
. front.....	35
. implementationCopier.....	36
. insertBack.....	36
. insertFront.....	35
. removeBack.....	36
. removeFront.....	35
. superpattern	
. . sequentialContainer.....	35
Deque.....	6
diff.....	23
doEnter.....	22
E	
element.....	19
elementPredicate.....	22
empty.....	20
emptyContainer.....	21
emptyContainerError.....	21
equal.....	20
extend.....	32
ExtensibleHashTable.....	32
. extend.....	32
. implementationCopier.....	32
. rehash.....	32
. superpattern	
. . hashTable.....	32
ExtensibleHashTable.....	5
F	
find.....	21
findIndexed.....	30
front.....	34, 35
G	
get.....	28
H	
has.....	20, 30
hashFunction.....	30
hashTable.....	29
. clear.....	29
. delete.....	30
. findIndexed.....	30
. has.....	30
. hashFunction.....	30
. . superpattern	
. . . indexValue.....	42
. implementationCopier.....	32
. implementationScanner.....	31
. index.....	30
. . superpattern	
. . . indexValue.....	42
. indexValue.....	29
. . subpatterns	
. . . hashFunction.....	30
. . . index.....	30
. . superpattern	
. . . integerValue.....	42
. insert.....	30

- . integerValue
 - . . subpatterns
 - . . . indexValue 29
 - . . . range 29
 - . . . rangeInitial 29
 - . range 29
 - . . superpattern
 - . . . integerValue 42
 - . rangeInitial 29
 - . . superpattern
 - . . . integerValue 42
 - . scanIndexed 30
 - . size 29
 - . statistics 31
 - . storage 32
 - . subpatterns
 - ExtensibleHashTable 32
 - . superpattern
 - . . collection 29
 - . theCellType 31
 - . theIndex 30
 - HashTable 5
 - head 37
- I**
- illegalCellReference 21
 - illegalSublistException 42
 - implementationCopier22, 24, 25, 27, 29, 32, 33, 34, 35, 36, 41, 42
 - implementationScanner22, 25, 27, 28, 31, 33, 35, 40, 42
 - index 30
 - indexValue 29
 - init 20
 - insert 23, 24, 25, 26, 30, 34
 - insertAfter 37
 - insertBack 36
 - insertBefore 37
 - insertFront 35
 - insertSublistAfter 42
 - insertSublistBefore 42
 - insertSubset 26
 - iterate 39
 - iterateFrom 40
 - iterateReverse 40
 - iterateReverseFrom 40
- L**
- last 37
 - less 27
 - list 36
 - . append 37
 - . at 38
 - . booleanValue
 - . . subpatterns
 - . . . cellPredicate 40
 - . cellPredicate 40
 - . . superpattern
 - . . . booleanValue 42
 - . clear 37
 - . concatenate 37
 - . delete 38
 - . deleteAfter 38
 - . deleteBefore 38
 - . head 37
 - . implementationCopier 41
 - . implementationScanner 40
 - . insertAfter 37
 - . insertBefore 37
 - . iterate 39
 - . iterateFrom 40
 - . iterateReverse 40
 - . iterateReverseFrom 40
 - . last 37
 - . locate 39
 - . preamble 37
 - . prepend 37
 - . scanFrom 39
 - . scanReverse 39
 - . scanReverseFrom 39
 - . size 37
 - . splitAfter 38
 - . splitBefore 38
 - . storage 41
 - subpatterns
 - recList 41
 - . superpattern
 - . . container 36
 - . tail 37
 - . theCellType 36
 - List 7
 - locate 39
 - more 22
 - moreOP 22
 - multiSet 24
 - . clear 24
 - . delete 25
 - . implementationCopier 25
 - . implementationScanner 25
 - . insert 24
 - . size 24
 - . storage 25
 - . subpatterns
 - set25
 - . superpattern
 - . . collection 24
 - . theCellType 25
 - MultiSet 2
- N**
- next 22
 - nextOP 22
- P**
- pop 33
 - preamble 37
 - prepend 37
 - prioQueue 34
 - . front 34
 - . implementationCopier 35
 - . implementationScanner 35
 - . insert 34
 - . remove 34
 - . scanPriority 35
 - . size 34
 - . superpattern

. . sequentialContainer	34	. clear.....	33
. theCellType	35	. implementationCopier	33
PrioQueue	6	. implementationScanner	33
push	33	. size.....	33
put	27	. storage	33
Q		. subpatterns	
queue	34	deque.....	35
. front	34	prioQueue.....	34
. implementationCopier.....	34	queue.....	34
. insert	34	stack.....	33
. remove.....	34	. superpattern	
. superpattern		. . container.....	33
. . sequentialContainer	34	. theCellType.....	33
Queue.....	6	SequentialContainer	6
quickSort	28	set25	
R		. implementationCopier	25
range.....	29	. insert.....	25
rangeInitial.....	29	. subpatterns	
recList	41	classificationSet.....	26
. Exception		. superpattern	
. . subpatterns		. . multiSet.....	25
. . illegalSublistException	42	Set	3
. illegalSublistException.....	42	shellSort.....	28
. . superpattern		size	20, 24, 26, 27, 29, 33, 34, 37
. . Exception.....	42	splitAfter.....	38
. implementationCopier.....	42	splitBefore.....	38
. implementationScanner.....	42	stack.....	33
. insertSublistAfter.....	42	. implementationCopier	33
. insertSublistBefore.....	42	. pop.....	33
. superpattern		. push.....	33
. . list.....	41	. superpattern	
. theCellType	41	. . sequentialContainer.....	33
RecList.....	8	. top.....	33
rehash.....	32	Stack	6
remove	34	start.....	22
removeBack	36	startOP.....	22
removeFront	35	statistics	31
S		storage	25, 29, 32, 33, 41
scan	20	subsets	26
scanFrom	39	superSet	26
scanIndexed.....	30	symDiff.....	24
scanPriority	35	T	
scanReverse.....	39	tail	37
scanReverseFrom	39	theCellType . 22, 24, 25, 28, 31, 33, 35, 36, 41	
scanUnclassified.....	26	theIndex	30
sect	24	top	33
sequentialContainer.....	33	U	
		union	23